Problem:
Score : 400 points

Problem Statement

There are N one-off jobs available. If you take the i-th job and complete it, you will earn the reward of B_i after A_i days from the day you do it.

You can take and complete at most one of these jobs in a day.

However, you cannot retake a job that you have already done.

Find the maximum total reward that you can earn no later than M days from today.

You can already start working today.

Constraints

All values in input are integers.

1 \leq N \leq 10^5

1 \leq M \leq 10^5

1 \leq A_i \leq 10^5

1 \leq B_i \leq 10^4

Input

Input is given from Standard Input in the following format:

N M
A_1 B_1
A_2 B_2
\vdots
A_N B_N

Output

Print the maximum total reward that you can earn no later than M days from today.

Sample Input 1

3 4
4 3
4 1
2 2

Sample Output 1

5

You can earn the total reward of 5 by taking the jobs as follows:

Take and complete the first job today. You will earn the reward of 3 after four days from today.

Take and complete the third job tomorrow. You will earn the reward of 2 after two days from tomorrow, that is, after three days from today.

Sample Input 2

5 3
1 2
1 3
1 4
2 1
2 3

Sample Output 2

10

Sample Input 3

1 1
2 1

Sample Output 3

0

Solution in OCaml (4.02.3) (judge verdict: Accepted)

open Scanf
open Printf

type heap = Nil | Node of int * heap * heap

let rec merge a b =
  match a, b with
  | a, Nil -> a
  | Nil, b -> b
  | Node (av, al, ar), Node (bv, bl, br) ->
    if av > bv then
      Node (av, (merge ar b), al)
    else
      Node (bv, (merge br a), bl)

let is_empty = function
  | Nil -> true
  | _ -> false

let push a v = merge a (Node (v, Nil, Nil))
let peek = function
  | Nil -> invalid_arg "nil"
  | Node (v, _, _) -> v

let pop = function
  | Nil -> invalid_arg "nil"
  | Node (_, l, r) -> merge l r


let () =
  let n, m = scanf " %d %d" (fun a b -> a, b) in
  let ab = Array.init n (fun _ -> scanf " %d %d" (fun a b -> a, b)) in
  Array.sort compare ab;
  let k = ref 0 in
  let q = ref Nil in
  let ans = ref 0 in
  for i = m - 1 downto 0 do
    while !k < n && i + fst ab.(!k) <= m do
      q := push !q (snd ab.(!k));
      incr k
    done;
    if not (is_empty !q) then begin
      ans := !ans + peek !q;
      q := pop !q
    end;
  done;
  printf "%d\n" !ans